=======DC_Controled_StateVariable_AC======




```
*          Y(S)/U(S)             =  K*(1+ s*b1 + s^2*b2)/( s^2 +a2*s +a1)
*          For Complex_Poles        1/(s^2 +a2*s +a1)  = 1/(s^2 +(wo/Q)*s +wo^2)
*          Note wo_rad/sec  =  Fc_Hz*2*PI
```

The direct mapping of DC voltages to transfer function constants can make filter construction easy.

```
*=========Create_Signal=================
VT         VT     0      DC      0    PWL( 0 0 100 100)
VFIN       FIN    0      DC      .7
BVIN       VIN    0      V       =    1*sin(6.283*V(FIN)*V(VT))
*=========Create_DC_Controls=================
VFC        FC     0      DC      1
VK         K      0      DC      1
VA1        A1     0      DC      1
VA2        A2     0      DC      1
VB0        B0     0      DC      1
VB1        B1     0      DC      1u
VB2        B2     0      DC      1u
XStateS    VIN    FC     K   A1  A2  B0  B1  B2  VOUT  HP  BP  LP  StateVS
```

The B0, B1, B2 terms control the feedforward, which in turn defines whether the output is a Highpass or Lowpass, etc.

Normally only the A2 terms get adjusted to change Q. Changing FC is a much cleaner way to adjust frequency.

```
.control
set        pensize = 2
*=========Start_Off_LowPass_Q_eq_1==========================
*AC        DECLin NUMDEC FSTART FSTOP
ac         dec     100    10m     100
plot       db(bp) db(hp) db(lp)         title Q_is_1
plot       db(vout) ph(vout)            title LowPassOut
*=========Make_Bandpass==========================
alter      VB0           dc = 1p
alter      VB1           dc = 1
ac         dec     100    10m     100
plot       db(vout) ph(vout)            title BandPassOut
*=========Make_Highpass==========================
alter      VB1           dc = 1p
alter      VB2           dc = 1
ac         dec     100    10m     100
plot       db(vout) ph(vout)            title HighPassOut
*=========Make_AllPass===========================
alter      VB0           dc = 1
alter      VB1           dc = -1
alter      VB2           dc = 1
ac         dec     100    10m     100
plot       db(vout) unwrap(ph(vout))    title AllPassOut
*=========Make_Q_eq_10===========================
alter      VA2           dc = .1
ac         dec     100    10m     100
plot       db(bp) db(hp) db(lp)         title Q_is_10
*=========Make_Q_eq_.7===========================
alter      VA2           dc = 1.414
ac         dec     100    10m     100
plot       db(bp) db(hp) db(lp)         title Q_is_.707
*=========Make_FC_eq_2===========================
alter      VFC           dc = 2
alter      VA2           dc = 1
*AC        DECLin NUMDEC FSTART FSTOP
ac         dec     100    10m     100
plot       db(bp) db(hp) db(lp)         title FC_is_2
.endc
```

This spice simulation shows that any type of filter can be defined by changing the DC Control voltages to this subcircuit.



For reference sake, the HP, BP, and LP nodes can be plotted to show what is happening inside the subcircuit.

```
*=========StateVariable_Cell_S=========================================
*
*
*         |\  K      HP   |\  b2 gmb2
*| IN |__|  \/\/\_________|  \/\/\_____________________________|OUT|
*|____|  |  /\/\/         |   /\/\/                         |  |___|
*         |/  gmin     |   |/        |\  b1 gmb1           |
*                      |          __|  \/\/\_______________|
*                      |         |  |  /\/\/               |
*                      |  Fc     |  |/     Fc              |
*         R1 =1        |  __|__  |       __|__   |\  b0    |    R2 =1
*     _/\  /\  /\___|_|  1/s  |_|______|  1/s  |___|  \/\/\_|_/\  /\  /\_
*    _|_ \/  \/    |  |______|  |  |  |______|  |   /\/\/     \/  \/  _|_
*    ///           |  s1     /  |  |    s2     |  |/  gmb0             ///
*                  |_____/\/\/  |_|BP           |
*                  |     \/\/\  |gma1           |
*                  |     -a2  \ |          /|   |LP
*                  |                     /\/\/ |_|
*                  |___________________\/\/\  |gma2
*                                      -a1  \|
.SUBCKT    StateVS VIN    FC   K   A1  A2  B0  B1  B2  VOUT  HP  BP  LP
```





The VOUT voltage is defined from the B0, B1, and B2 values. These values define the filter type.





For ALL pass filters, B0 and B2 are one and B1 is minus one.





Independently, adjusting Q is as simple as adjusting A2.
In this case Q is set at 10, and then set at 0.7.



The cleanest way to independently adjust frequency is to adjust FC. In this case FC has been set from 1 to 2.

===========Full_Netlist_For_Copy_Paste=======================

```
DC_Controled_StateVariable_AC
*                              ___
*                        _____| b2|_____________________
*                       |     |___|     ___             |
*                       |          ____| b1|__________  |
*                       |Highpass |    |___|          | |
* U(S)        ___   ..  |         | .                 V_V
*     ___    / _ \  y(t)|  _____  | y(t) _____ y(t)  / _ \
*  _\| k |_\/ \   \_____|_| 1/s |_|_____| 1/s |____\/ \   \___\ Y(S)
*   /|___| /\ /_  /       |_____|   |   |_____||   /\ /_  /   /
*            \___/                  |          |     \___/
*             ^ ^    _____ Bandpass |          |
*             | |___| -a2 |_________|          |  b0 = 1
*             |     |_____|  _____     Lowpass |
*             |_____________| -a1 |____________|   w0 = sqrt(a1)
*                           |_____|                Q  = w0/a2
*
*         Y(S)/U(S)  =   K*(1+ s*b1 + s^2*b2)/( s^2 +a2*s +a1)
*         Complex_Poles  1/(s^2 +a2*s +a1)  = 1/(s^2 +(wo/Q)*s +wo^2)
*
*=========Create_Signal==================
VIN       VIN    0      DC     0   AC   1
VFC       FC     0      DC     1
VK        K      0      DC     1
VA1       A1     0      DC     1
VA2       A2     0      DC     1
VB0       B0     0      DC     1
VB1       B1     0      DC     1p
VB2       B2     0      DC     1p
XStateS   VIN    FC     K   A1  A2  B0  B1  B2  VOUT  HP  BP  LP  StateVS
.control
set       pensize = 2
*=========Start_Off_LowPass_Q_eq_1===========================
*AC       DECLin NUMDEC FSTART FSTOP
ac        dec    100    10m     100
plot      db(bp) db(hp) db(lp)        title Q_is_1
plot      db(vout) ph(vout)           title LowPassOut
*=========Make_Bandpass===========================
alter     VB0           dc = 1p
alter     VB1           dc = 1
ac        dec    100    10m     100
plot      db(vout) ph(vout)           title BandPassOut
*=========Make_Highpass===========================
alter     VB1           dc = 1p
alter     VB2           dc = 1
```

```
ac        dec    100    10m     100
plot      db(vout) ph(vout)            title HighPassOut
*=========Make_AllPass===========================
alter     VB0           dc = 1
alter     VB1           dc = -1
alter     VB2           dc = 1
ac        dec    100    10m     100
plot      db(vout) unwrap(ph(vout))    title AllPassOut
*=========Make_Q_eq_10===========================
alter     VA2           dc = .1
ac        dec    100    10m     100
plot      db(bp) db(hp) db(lp)         title Q_is_10
*=========Make_Q_eq_.7===========================
alter     VA2           dc = 1.414
ac        dec    100    10m     100
plot      db(bp) db(hp) db(lp)         title Q_is_.707
*=========Make_FC_eq_2===========================
alter     VFC           dc = 2
alter     VA2           dc = 1
*AC       DECLin NUMDEC FSTART FSTOP
ac        dec    100    10m     100
plot      db(bp) db(hp) db(lp)         title FC_is_2
.endc
*=========StateVariable_Cell_S==========================================
*
*
*  ___   |\  K      HP   |\  b2 gmb2                        ___
*| IN |__| \/\/\_________| \/\/\___________________________|OUT|
*|___|   | /\/\/     |   | /\/\/                        |  |___|
*        |/  gmin    |   |/      |\  b1 gmb1            |
*                    |         __| \/\/\________________|
*                    |        |  | /\/\/                |
*                    |  Fscale|  |/   Fscale            |
*     R1 =1          |   __|__|       __|__   |\  b0    |     R2 =1
*   _/\  /\  /\___|_|  1/s  |_|______|  1/s  |___| \/\/\_|_/\  /\  /\_
*  _|_ \/  \/     |  |______|   |    |______| |  | /\/\/    \/  \/  _|_
*  ///            |  s1       /|   |    s2    |  |/  gmb0            ///
*                 |_____/\/\/ |_|BP          |
*                       \/\/\ |gma1          |
*                       -a2  \|          /|  |LP
*                 |_______________/\/\/ |_|
*                                 \/\/\ |gma2
*                                 -a1  \|
.SUBCKT   StateVS VIN    FC   K   A1  A2  B0  B1  B2  VOUT  HP  BP  LP
 R1        HP     0      1
 R2        OUT    0      1
 Bgmin     HP     0      I  = -V(VIN)*V(K)*1
 Bgma1     HP     0      I  =  V(LP)*V(A1)
 Bgma2     HP     0      I  =  (V(BP))*V(A2)
 Bgmb0     OUT    0      I  = -V(LP)*V(B0)
 Bgmb1     OUT    0      I  = -V(BP)*V(B1)
 Bgmb2     OUT    0      I  = -V(HP)*V(B2)
 XS1block  HP     BP     FC    Sblock
 XS2block  BP     LP     FC    Sblock
 BOUT      VOUT   0      V  =  V(OUT)
.ENDS     StateVS
*=========S_BLOCK========================================
*                            _______
*         IN   RS=1/FC    _|_       |
*       |\                          |
* ___   |                 ___       |
*|VIN|__| \_/\  /\  /\____|____|\   |  ___
*|___|  | /   \/  \/   VNEG    | \_|_|OUT|
*       |/Gain=-1          ____| /    |___|
*                         _|_  |/
*   For FC = 1Hz          ///
*   RS = 1 Ohm                 CS=1uF/2Pi
*   Xc = 1 Ohm
.SUBCKT   Sblock VIN    OUT    FC
 Bbuf     IN     0      V =    -V(VIN)
 BRS      IN     VNEG   I =    (V(IN)-V(VNEG))*V(FC)
 Cs       VNEG   OUT    .159
 BSOUT    OUT    0      V =    -V(VNEG)*1000
.ENDS     Sblock
.end
```


6.7.11_10.39AM
dsauersanjose@aol.com
Don Sauer